[Nexus 7 Original Development] osm0sis:
For someone who's asking for help you sure are quick to dismiss it.

Tell me, how unsafe does this look: :rolleyes:

ui_print("");
ui_print("Kernel Emergency Reset Script");
ui_print("by osm0sis @ xda-developers");
show_progress(1.34, 0);

ui_print("");
ui_print("Mounting...");
run_program("/sbin/busybox", "mount", "/system");
run_program("/sbin/busybox", "mount", "/data");
run_program("/sbin/busybox", "mount", "/cache");
set_progress(0.2);

ui_print("");
ui_print("Disabling init.d and userinit.d scripts...");
run_program("/sbin/busybox", "mv", "/system/etc/init.d", "/tmp/off");
run_program("/sbin/busybox", "mkdir", "/system/etc/init.d");
run_program("/sbin/busybox", "mv", "/tmp/off", "/system/etc/init.d/off");
set_perm_recursive(0, 0, 0755, 0755, "/system/etc/init.d");

ifelse(run_program("/sbin/sh", "-c", "test -e /data/local/userinit.d") == 0,
(run_program("/sbin/busybox", "mv", "/data/local/userinit.d", "/tmp/off");
run_program("/sbin/busybox", "mkdir", "/data/local/userinit.d");
run_program("/sbin/busybox", "mv", "/tmp/off", "/data/local/userinit.d/off");
set_perm_recursive(0, 0, 0755, 0755, "/data/local/userinit.d");
)
);
set_progress(0.4);

ui_print("");
ui_print("Deleting Kernel App Settings...");
delete("/data/data/aperture.ezekeel.experience/aperture.ezekeel.experience_preferences.xml");
delete("/data/data/aperture.ezekeel.experiencepro/aperture.ezekeel.experiencepro_preferences.xml");
delete("/data/data/aperture.ezekeel.gladoscontrol/shared_prefs/aperture.ezekeel.gladoscontrol_preferences.xml");
delete("/data/data/com.bigeyes0x0.trickstermod/shared_prefs/settings.xml");
delete_recursive("/data/data/com.derkernel.tkt/shared_prefs");
delete("/data/data/com.franco.kernel/shared_prefs/com.franco.kernel_preferences.xml", "/data/data/com.franco.kernel/shared_prefs/custom_io_values.xml");
delete_recursive("/data/data/com.jrummy.liberty.toolbox/shared_prefs");
delete_recursive("/data/data/com.jrummy.liberty.toolboxpro/shared_prefs");
delete("/data/data/com.mhuang.overclocking/shared_prefs/setcpu.xml");
delete_recursive("/data/data/com.teamkang.fauxclock/shared_prefs");

ui_print("");
ifelse(run_program("/sbin/sh", "-c", "test -e /data/data/com.brewcrewfoo.performance") == 0,
(ui_print("Deleting ROM Performance App Settings...");
delete("/data/data/com.brewcrewfoo.performance/shared_prefs/com.brewcrewfoo.performance_preferences.xml");
),
(ui_print("Note: If your ROM has a Performance/Tweaking");
ui_print(" App built-in you should also manually");
ui_print(" disable any Set On Boot options it has.");
)
);
set_progress(0.6);

ui_print("");
ui_print("Wiping dalvik-cache...");
delete_recursive("/data/dalvik-cache");
set_progress(0.8);

ui_print("");
ui_print("Wiping cache...");
delete_recursive("/cache");
set_progress(1.0);

ui_print("");
ui_print("Unmounting...");
unmount("/system");
unmount("/data");
unmount("/cache");
set_progress(1.2);

ui_print("");
ui_print("Done!");
ui_print("");
ui_print("Hopefully that fixed whatever you broke.");
ui_print("Please reboot.");
set_progress(1.34);

It really doesn't do much, but the fact remains you're the only one having problems with r63 and CM, so if I were you, I'd be trying anything.